Volumes That You Can Use

Note: You may notice differences between this Maple worksheet and the equivalent Mathematica notebook. These differences were introduced to preserve the content of these modules and were necessary because of major functional differences between Maple and Mathematica.

Introduction

OBJECTIVE: Use the concept of volume to solve practical problems involving rain catchers and satellite dishes.

How can you measure precipitation if you collect rain in a container that is narrower at the bottom than at the top? At what angle must a satellite dish be tilted for rain not to collect in it?

Technology Guidelines

NOTE: If you have just finished a worksheet, **restart** *Maple* before executing a new worksheet.
TO OPEN SECTIONS,
Click on the **PLUS** sign at the left hand side of the screen *or* select **Expand All Sections** from the **View** drop down menu.

TO STOP AN EXECUTION
Click on **STOP** button from the toolbar.

ORDER OF EXECUTION
Execute commands in the order given. Do not skip any *Maple* Input lines within a given worksheet

Alternatively, you can execute the entire worksheet by selecting the **Execute Worksheet** command from the **Edit** drop down menu.

SAVING WORKSHEETS.

You can save anytime to any directory you choose, and it is wise to save often.
EXPERIENCING MAJOR PROBLEMS
Save if appropriate, and then shut down *Maple* and start it up again.

Part I: A Hemispherical Rain Gauge

End of Chapter Exercises, Variation of Exercise 25

Suppose you are trying to determine how to use the height of water in a hemispherical bowl of radius and depth 10 inches to measure the amount of rain that has fallen. Why do we use the volume of a cylinder of radius 10 inches filled to a particular height (k)? What does that have to do with the volume of water that will collect in the bowl? The key idea is that the volume in each is the same; the only difference is in the height of the water in the two different containers.

We will write our formulas using a general radius (ra), so that we can go back and change it out for hemispherical bowls with different radii. Since the bowl is considered to be sitting on top of the *xy*-plane with the axis of symmetry through the *z* axis, it is easiest to use cylindrical coordinates. The equation of the sphere is $r^2 + (z - rad)^2 = rad^2$.

```
> restart;
```

```
> volcylinderht:= Pi*ra^2*k:
  volbowlht:=int(int(int(r,z=(ra-sqrt(ra^2-r^2))..h),r=0..sqrt(2*h*ra-h^2)),theta=0..2*Pi):
  print(`volume in cylinder filled to height k is `, volcylinderht);
  print(`volume in bowl filled to height h is `, volbowlht);
```

$$\textit{volume in cylinder filled to height k is }, \pi\, ra^2\, k$$

$$\textit{volume in bowl filled to height h is }, \frac{2}{3}\,\mathrm{csgn}(ra)\, ra^3\, \pi + 2\,\mathrm{csgn}(-h+ra)\, ra^2\, h\, \pi - 2\,\mathrm{csgn}(-h+ra)$$
$$+\frac{2}{3}\,\mathrm{csgn}(-h+ra)\, h^3\, \pi - \frac{2}{3}\,\mathrm{csgn}(-h+ra)\, ra^3\, \pi + 3\, h^2\, ra\, \pi - 2\, ra^2\, h\, \pi - h^3\, \pi$$

If there is 1 inch of rain in the 10-inch radius cylinder, we will determine the height of water in the bowl. Why do we equate the volume in the cylinder to the volume in the bowl?

```
> volc:=subs({ra=10, k=1},volcylinderht):
  volb:=subs(ra=10,volbowlht):
  one:=fsolve(volb=volc,h);
```

$$one := 3.355496675$$

```
> print(`If there is one inch of rain in the 10 - inch radius cylinder, the hemisphere will be fil
  the `, one, `inch level`);
```

If there is one inch of rain in the 10 - inch radius cylinder, the hemisphere will be filled to the , 3.3
inch level

What if there are 3 inches of water in the bowl? How many inches of water would this produce in the cylinder?

```
> volc:=subs(ra=10,volcylinderht):
  volb:=subs({ra=10, h=3},volbowlht):
  three:=fsolve(volb=volc):
  print(`If there are three inches of rain in the bowl, the actual precipitation level is `, three,
  `inches`);
```

If there are three inches of rain in the bowl, the actual precipitation level is , 0.8100000001, *inche*

Because of multiple roots, it is somewhat difficult to solve explicitly for the height in the bowl as a function of the actual amount of rain (height of water in the cylinder). However, it is easy to determine the relationship the other way around; that is, if we know the height of water in the bowl, we can easily find the amount of precipitation by dividing the volume of water in the bowl by $\pi\, rad^2$, which is the cross-sectional area of the cylinder with the same size top. That

computes the actual precipitation amount.

We do this for a general radius and look at some tables of values associating the actual precipitation and the height of the water in the bowl corresponding to different values of the radius. The formulas below simply show that the height in the cylinder, representing actual precipitation, is simply the volume of collected water divided by the cross-sectional area of the cylinder $\pi\, r^2$.

```
> precip:=volbowlht/(ra^2*Pi);
```

$$precip := \left(\frac{2}{3}\,\mathrm{csgn}(ra)\,ra^3\,\pi + 2\,\mathrm{csgn}(-h+ra)\,ra^2\,h\,\pi - 2\,\mathrm{csgn}(-h+ra)\,h^2\,ra\,\pi + \frac{2}{3}\,\mathrm{csgn}(-h+ra\right.$$
$$\left. - \frac{2}{3}\,\mathrm{csgn}(-h+ra)\,ra^3\,\pi + 3\,h^2\,ra\,\pi - 2\,ra^2\,h\,\pi - h^3\,\pi\right) \Big/ \left(ra^2\,\pi\right)$$

We can apply these formulas to create a list of values that associate the height of water in the bowl with the actual precipitation.

```
> Digits:=20:
  precip1:=subs(ra=10, precip):
  seq([0.5*i, subs(h=0.5*i, precip1)], i=1..20):
  preciplist1:=evalf(%):
  [`10" bowl level    `,`    actual precip.`];
  matrix([preciplist1]);
  p1:=plots[pointplot]({preciplist1}, labels=["height","precipitation"], color=blue):

   #print(p1);
```

[*10" bowl level* , *actual precip.*]

0.5	0.02458333333333333300
1.0	0.09666666666666666700
1.5	0.2137500000000000000
2.0	0.3733333333333333330
2.5	0.5729166666666666670
3.0	0.8100000000000000000
3.5	1.082083333333333333
4.0	1.386666666666666667
4.5	1.721250000000000000
5.0	2.083333333333333333
5.5	2.470416666666666667
6.0	2.880000000000000000
6.5	3.309583333333333333
7.0	3.756666666666666670
7.5	4.218750000000000000
8.0	4.693333333333333340
8.5	5.177916666666666670
9.0	5.670000000000000000
9.5	6.167083333333333330
10.0	6.666666666666666670

```
> precip2:=subs(ra=15, precip):
seq([0.5*i, subs(h=0.5*i, precip2)], i=1..20):
preciplist2:=evalf(%):
[`15" bowl level   `,`   actual precip.`];
matrix([preciplist2]);
p2:=plots[pointplot]({preciplist2}, labels=["height","precipitation"], color=green):
#print(p2);
```

[*15" bowl level* , *actual precip.*]

0.5	0.01648148148148148133
1.0	0.06518518518518518533
1.5	0.14500000000000000000
2.0	0.2548148148148148146
2.5	0.3935185185185185186
3.0	0.5600000000000000000
3.5	0.7531481481481481480
4.0	0.9718518518518518518
4.5	1.2150000000000000000
5.0	1.4814814814814814815
5.5	1.7701851851851851852
6.0	2.0800000000000000000
6.5	2.4098148148148148148
7.0	2.7585185185185185187
7.5	3.1250000000000000000
8.0	3.5081481481481481480
8.5	3.9068518518518518520
9.0	4.3200000000000000000
9.5	4.7464814814814814813
10.0	5.1851851851851851853

```
> precip3:=subs(ra=20, precip):
seq([0.5*i, subs(h=0.5*i, precip3)], i=1..20):
preciplist3:=evalf(%):
p3:=plots[pointplot]({preciplist3}, labels=["height","precipitation"], color=red):
#print(p3);
[`20" bowl level   `,`   actual precip.`];
matrix([preciplist3]);
```

[*20" bowl level* , *actual precip.*]

$$\begin{bmatrix} 0.5 & 0.012395833333333333250 \\ 1.0 & 0.049166666666666666750 \\ 1.5 & 0.10968750000000000000 \\ 2.0 & 0.19333333333333333325 \\ 2.5 & 0.29947916666666666675 \\ 3.0 & 0.42750000000000000000 \\ 3.5 & 0.57677083333333333325 \\ 4.0 & 0.74666666666666666675 \\ 4.5 & 0.93656250000000000000 \\ 5.0 & 1.1458333333333333332 \\ 5.5 & 1.3738541666666666668 \\ 6.0 & 1.6200000000000000000 \\ 6.5 & 1.8836458333333333332 \\ 7.0 & 2.1641666666666666668 \\ 7.5 & 2.4609375000000000000 \\ 8.0 & 2.7733333333333333332 \\ 8.5 & 3.1007291666666666668 \\ 9.0 & 3.4425000000000000000 \\ 9.5 & 3.7980208333333333332 \\ 10.0 & 4.1666666666666666668 \end{bmatrix}$$

```
> print(plots[display]({p1,p2,p3}));
```



Which plot goes with which radius?

You Try It: Part I

Now, select a different shaped rain-collector, and determine the relationship between the height of water in your collector and the actual precipitation. Suppose that the bowl is parabolic, say with the equation $z = .5\,r^2$,with maximum radius 10 inches.

```
> Digits:=8:
  volofbowlht:=int(int(int(r,z=.5*r^2..h),r=0..sqrt(2*h)),theta=0..2*Pi);
```

$$volofbowlht := 3.1415927\,h^2$$

You can enter ra = 10 at the start and replace the **volofbowlht** entry with the above and proceed. The relationship is much simpler here. Give the precipitation level for height *h* in your bowl as follows.

```
> ra:=10:
  newprecip:=simplify(volofbowlht/(ra^2)*Pi);
```

$$newprecip := 0.098696047\,h^2$$

Find the precipiation level if the height in the bowl is 2.5 inches.

```
> subs(h=2.5, newprecip);
```

$$0.61685029$$

Repeat these steps for containers having different shapes.

Part II: Water in a Satellite Dish

End of Chapter Exercises, Exercise 26 and extension

As we tilt the bowl, it is convenient to use rectangular coordinates and picture the tilted bowl upright and show the water as tilted. We will let the bowl be centered on the z axis and rest on top of the xy-plane. The equation of the plane that represents the water level can be written as $z = m\, y + b$. In each case, since the water level plane must pass through the coordinate points $z = 0.5$ when $y = 1$, the intercept b can be written as $0.5 - m$.

```
> xyzbowl:=(x,y)->.5*(x^2+y^2);
  xyzplane:=(y,m)->m*y+(.5-m);
  plotbowl:=plot({xyzbowl(0,y), xyzplane(y,0)}, y=-1..1, color=[blue, cyan]):
  plot1:=plot(xyzplane(y,.25), y=-.5..1, color=green):
  plot2:=plot(xyzplane(y,.5),y=0..1, color=red):
  print(plots[display]({plotbowl, plot1, plot2}));
```

$$xyzbowl := (x, y) \to 0.5\,(x^2 + y^2)$$

$$xyzplane := (y, m) \to m\, y + 0.5 - m$$



Let us begin by drawing some scenarios as the slope of the plane (value of m) goes from 0 to .75. We can integrate to find the volume in each case first by finding the boundary curve for x and y along which the plane and the paraboloid intersect and then by leaving our integral written as a function of the slope of the water level plane.

```
> sol:=solve(xyzplane(y,m)=xyzbowl(x,y),y);
  sol1:=unapply(sol[1], (x,m));
```

```
sol2:=unapply(sol[2], (x,m));
```

$$sol := m+\sqrt{m^2+1.-2.\,m-1.\,x^2},\ m-1.\,\sqrt{m^2+1.-2.\,m-1.\,x^2}$$

$$sol1 := (x,m) \rightarrow m+\sqrt{m^2+1.-2.\,m-1.\,x^2}$$

$$sol2 := (x,m) \rightarrow m-1.\,\sqrt{m^2+1.-2.\,m-1.\,x^2}$$

Note that the equation of intersection of the plane and the paraboloid is a circle, centered at $x = 0$ and y = m and with radius 1 - m, so we will let z go from the paraboloid to the plane and integrate x and y over the appropriate circle.

```
> int(int(int(1, z=xyzbowl(x,y)..xyzplane(y,m)), y=sol[1]..sol[2]), x=(m-1)..(1-m)):
  vol:=unapply(%,m);
```

$$vol := m \rightarrow \int_{m-1}^{1-m} 0.50000000\,m\left(\left(m-1.\,\sqrt{m^2+1.-2.\,m-1.\,x^2}\right)^2-1.\,\left(m+\sqrt{m^2+1.-2.\,\ldots\right.\right.$$
$$-1.\,\sqrt{m^2+1.-2.\,m-1.\,x^2}+2.\,m\sqrt{m^2+1.-2.\,m-1.\,x^2}+x^2\sqrt{m^2+1.-2.\,m-1.\,x^2}$$
$$-0.16666667\left(m-1.\,\sqrt{m^2+1.-2.\,m-1.\,x^2}\right)^3+0.16666667\left(m+\sqrt{m^2+1.-2.\,m-1.\,x^2}\right.$$

Now, let's visualize the results for a few values of m. We first read in a package to help us plot. Then we let the slope take on the values of .25, 1, and .75, each time evaluating the volume of water in the satellite.

```
> plot({xyzplane(y,0),xyzbowl(0,y)}, y=-1..1, labels=[y,z], title="Slope of Water is 0");
  print(`When the slope is 0, the volume of the water is `, abs(vol(0)),`cubic meters.`);
  plot({xyzplane(y,.25),xyzbowl(0,y)}, y=-.5..1, labels=[y,z], title=`Slope of Water is .25`);
  print(`When the slope is .25, the volume of the water is `, abs(vol(0.25)),`cubic meters.`);
  plot({xyzplane(y,.5),xyzbowl(0,y)}, y=0..1, labels=[y,z], title="Slope of Water is .5");
  print(`When the slope is .5, the volume of the water is `, abs(vol(.5)),`cubic meters.`);
  p3:=plot({xyzplane(y,.75),xyzbowl(0,y)}, y=.5..1, labels=[y,z], title="Slope of Water is .75")
  p4:=plot({xyzbowl(0,y)}, y=0...1, labels=[y,z], color=green):
  print(plots[display]({p3,p4}));
  print(`When the slope is .75, the volume of the water is `, abs(vol(.75)),`cubic meters.`);
```



When the slope is 0, the volume of the water is , 0.78539816, *cubic meters.*



When the slope is .25, the volume of the water is , 0.24850488, *cubic meters.*



When the slope is .5, the volume of the water is , 0.049087383, *cubic meters.*



When the slope is .75, the volume of the water is , 0.0030679604, *cubic meters.*

As you recognize that the slope of the water level represents the tangent of the angle through which the bowl is tilted, you will also notice that the volume is approaching 0 as m is approaching 1, and, hence, as the angle of tilt is approaching 45°.

```
> vol(1);
```

0.

>

You Try It: Part II

Check out satellite dishes that are installed at residences or at communication facilities. See if their shape is parabolic, and find out what their tilt is. Does it match up to what you expected?